3-D GRAPHICS! Finally! Real Graphics!

by Dale A. Keller

I WANTED A PROGRAM which could easily read a data file from tape so I could produce several graphics representations without changing data statements in a program, could rotate an object about more than one axis at a time, could change distance from the object (a zoom lens effect), and could display a continuous rotation without stopping to calculate between each position. The program should also have a fast, smooth display, and be easy to operate. I also wanted hidden line removal, but found the memory limitations (both the computer's and mine) to fall short of the goal.

I wrote a program that would meet my requirements. This program reads a data file from tape, and then calculates and displays an object. To get the data file on tape, I wrote a drawing editor. I was surprised to find that it was about twice as long as the rotation program after I included everything I needed in the editor. To use the editor, we must define an object as a list of three-dimensional line coordinates.

Defining 3-D Objects

An easy way to define the coordinates of a three-dimensional object is to draw side, front, and top views of the object on graph paper, then make a list of the X, Y, and Z coordinates of the two endpoints of each line in the object.

For example, let's define the coordinates of a simple object: a cabin. First draw the side, front, and top views of the cabin on graph paper (see Figure 1). Then list coordinates along the edges of the graph paper, counting from 0, 0, 0 in the very center of the cabin. Next label the X axis along the length of the cabin, the Y axis across the width, and label the Z axis vertically.

We are now ready to make our list. We can start with any line, but we must keep track of those we have finished. Do this by numbering each line or working from one end of the object to the other. For this demonstration, I have numbered the lines. For the line numbered one, the coor-



Figure 1. Defining Cabin Coordinates

dinates are: X = 50, Y = 35, Z = −40 to X = −50, Y = 35, Z = −40. Continue the list until both end coordinates of all the lines have been defined (see Figure 2). On complex objects, it also helps to label the lines in the list (for instance, lines 1 through 4 are the west wall).

Line No.	From X	From Y	From Z	To X	To Y	To Z	
1	50,	35,	−40	−50,	35,	−40	
2	−50,	35,	−40	−50,	35,	25	West
3	50,	35,	−40	50,	35,	25	Wall
4	50,	35,	25	−50,	35,	25	
5	50,	−35,	−40	−50,	−35,	−40	
6	−50,	−35,	−40	−50,	−35,	25	East
7	50,	−35,	−40	50,	−35,	25	Wall
8	50,	−35,	25	−50,	−35,	25	
9	−50,	35,	−40	−50,	−35,	−40	
10	50,	35,	−40	50,	−35,	−40	Ends
11	−50,	35,	25	−50,	−35,	25	
12	50,	35,	25	50,	−35,	25	
13	−50,	−35,	25	−50,	0,	40	
14	−50,	35,	25	−50,	0,	40	
15	50,	−35,	25	50,	0,	40	Roof
16	50,	35,	25	50,	0,	40	
17	−50,	0,	40	50,	0,	40	

Figure 2. Line List For the Cabin

Using the Editor

Now our object has been defined as a set of coordinates, we must make a tape file of this list. Type in and CSAVE Program Listing 1. Then type: PCLEAR1:CLEAR 20,16383. This clears the graphics memory for use by Basic,and sets the upper limit of memory to 16383 bytes. If you try to run the program without first entering these commands, Basic will execute the first line and then stop. Running the program again will work fine.

The first screen you will see is the main menu. Since we are entering a new picture, hit 1 and Enter. Now you can enter the coordinates from Figure 2, first typing the first three numbers (in the From column) separated by commas, hitting Enter, then the second three numbers (in the To column) separated by commas. When all the lines have been entered, hit 999 followed by two commas, and Enter. This tells the editor you are finished entering new lines and it will return to the main menu.

To check your entries, hit 2, for Edit. When the Edit menu appears, hit 4 for List. Check each entry carefully, writing down any line numbers in error. If you need to make any corrections when the Edit menu reappears, hit the appropriate number and the line number if needed. Lines are changed or added one at a time by typing the correct coordinates.

When you have a correct listing, return to the main menu (hit 5). Now hit 3 to save your listing to tape. The screen will say WORKING.... Don't panic, it is not saving yet. It is storing the data in high memory, from where it will CSAVEM the program. When the computer is ready, it will ask for a drawing name. Enter any appropriate name (for instance: CABIN). When it is finished saving, return to the main menu and load the file you just saved. List it and check that everything is correct.

Displaying an Object

Type in and CSAVE Program Listing 2. Then type: PCLEAR8:CLEAR10,16383. If you run this program without first entering these commands, the program will run until just before the display and then quit.

The program first loads your data file from tape into high memory (Lines 40 – 80). It then finds how many lines are in the object (variable LC for Line Count) and tells you how many steps you may have according to memory limits (Line 90). See Figure 3 for typical number of steps.

Line 110 asks for the position for the starting view of the object. The distance is your distance from the object. The angle around is your horizontal angle in degrees from the X axis. The elevation angle is your vertical angle in degrees from the Z axis (see Figure 4). The first time through (with the cabin drawing) use a distance 1 of 3000, and angle around 1 of 90 degrees, an elevation angle 1 of 75 degrees, a distance 2 of 1500, an angle around 2 of 160 degrees, and an elevation angle 2 of 45 degrees. Line 150 asks for the number of steps. For the first time try only five steps.

Lines 160 through 320 get the coordinates of the object from high memory (addresses 16385 – 17920) and calculate their positions for each of your points of view. This data is stored in high memory (addresses 17921 – 32767). This process is very slow since the computer has to handle so many numbers. To calculate five positions for the 17 lines in the cabin it will take about 25 seconds. For more complex objects this can take up to 16½ minutes.

Lines	Steps Available	Computing Time (for 14 steps)
256	14	16.5 minutes
128	28	8.25 minutes
64	57	248 seconds
32	115	124 seconds
17	218	67 seconds

Figure 3. Time and Steps

Program Listing 1. 3D Editor

```
10 CLEAR20,16383:PCLEAR1
20 '3D EDITOR by Dale A. Keller
40 DIM A(1536)
50 CLS:PRINT"NEW PICTURE (1)":PR
INT"EDIT OLD PICTURE (2)":PRINT"
SAVE (3)":PRINT"LOAD (4)":INPUT
A
60 ON A GOTO 80,210,650,800
70 GOTO 50
80 'DRAW NEW PICTURE
90 CLS:PRINT"NEW PICTURE":PRINT"
HIT 999 TO EXIT":PRINT
```

more



Figure 4. Angle and Distance Measurements

When the computing is finished, the computer will ask if you are ready and play a short tune (make sure the volume is turned up on the TV so you can hear it from the refrigerator. Like I said, some of these rotations take a long time to compute).

Lines 330 and on draw the displays on alternating screens. For simple objects this is a fairly quick process. Complex objects are slower, but you still appear to be moving around the object. When the display stops, hitting any key will take you to a short menu in Line 470. The rotation can be viewed again immediately since it has already been figured, or you can rotate the same object through different angles.

There are some limitations to this program. You may only use up to 256 lines. The object should be centered on the coordinates 0, 0, 0 to rotate smoothly on its center. Lines must have coordinates in the range of −128 to 127. There is no line clipping routine, so lines must be within the graphics screen dimensions. If you use too short a distance, values of less than 0 or greater than 255 will be computed and the computer will try to POKE them somewhere, resulting in a Function Call Error. Using a greater distance will remedy this error.

The variable SC in Line 20 is a scale value. This compensates for the screen aspect ratio (256 by 192 pixels) and your TV aspect ratio (approximately 4 to 3). This should be changed to fit your TV. To do this, type in Listing 3 and run the program. Then measure the width and height of the square with a ruler. Use the formula SV = height/width to find SV.

When you get tired of the Cabin, try some other objects — cubes, pyramids, etc. Try rotating two objects around each other. Figure 5 is the coordinate list for a

Figure 5. Line List For a Space Shuttle

Line Number	From			To			Description
1	−44,	0,	14	55,	0,	15	Split Between Doors
2	55,	0,	15	60,	0,	10	Top Center Over Windows
3	60,	0,	10	83,	0,	0	
4	83,	0,	0	86,	0,	−3	
5	86,	0,	−3	85,	0,	−6	Nose (center)
6	85,	0,	−6	77,	0,	−10	
7	77,	0,	−10	45,	0,	−12	
8	−44,	0,	14	−45,	0,	19	
9	−45,	0,	19	−82,	0,	−52	
10	−82,	0,	52	−92,	0,	52	Tail
11	−92,	0,	52	−94,	0,	51	
12	−94,	0,	51	−80,	0,	22	
13	−80,	0,	22	−70,	0,	18	
14	−68,	−16,	−13	−61,	−58,	−9	
15	−61,	−58,	−9	−54,	−58,	−9	
16	−54,	−58,	−9	−42,	−56,	−10	Right Wing
17	−42,	−56,	−10	−13,	−26,	−12	
18	−13,	−26,	−12	47,	−13,	−10	
19	−68,	16,	−13	−61,	58,	−9	
20	−61,	58,	−9,	−54,	58,	−9	
21	−54,	58,	−9,	−42,	55,	−10	Left Wing
22	−42,	55,	−10	−13,	27,	−12	
23	−13,	27,	−12	47,	13,	−10	
24	45,	−13,	−14	45,	−13,	9	
25	45,	−13,	9	45,	−9,	13	
26	45,	−9,	13	45,	−4,	15	Front Body Arch
27	45,	−4,	15	45,	4,	15	
28	45,	4,	15	45,	9	13	
29	45,	9,	13	45,	13,	9	
30	45,	13,	9	45,	13,	−14	
31	−45,	−13,	−14	−45,	−13,	9	
32	−45,	−13,	9	−45,	−9,	13	
33	−45,	−9,	13	−45,	−4,	15	
34	−45,	−4,	15	−45,	4,	15	Rear Body Arch
35	−45,	4,	15	−45,	9,	13	
36	−45,	9,	13	−45,	13,	9	
37	−45,	13,	9	−45,	13,	−14	
38	45,	13,	−4	48,	13,	−4	
39	48,	13,	−4	78,	7,	−4	Left Side of Nose
40	78,	7,	−4	85,	3,	−4	
41	85,	3,	−4	86,	0,	−4	
42	45,	−13,	−4	48,	−13,	−4	
43	48,	−13,	−4	78,	−7,	−4	Right Side of Nose
44	78,	−7,	−4	85,	−3,	−4	
45	85,	−3,	−4	86,	0,	−4	
46	45,	13,	−12	−42,	13,	−15	Left Wing To Body Joint
47	−42,	13,	−15	−72,	13,	−14	
48	45,	−13,	−12	−42,	−13,	−15	Right Wing To Body Joint
49	−42,	−13,	−15	−72,	−13,	−14	
50	45,	13,	−12	45,	−13,	−12	Front Lower Body
51	−72,	13,	−14	−72,	−13,	−14	Rear Lower Body
52	−72,	14,	−14	−84,	14,	−13	Left Side Rear Flap
53	−84,	14,	−13	−72,	14,	−12	
54	−72,	−14,	−14	−84,	−14,	−13	Right Side Rear Flap
55	−84,	−14,	−13	−72,	−14,	−12	
56	−84,	14,	−13	−84,	−14,	−13	Rear Edge Of Flap

more

Space Shuttle. It takes some time to compute, but it's worth it.

I left the rotational math out of this article because there are many books and articles already describing it in detail. (See the References at the end of this article). The formulas used in this program came from *Microcomputer Graphics* by Roy E. Meyers. ■ ■ ■

Line Number	From			To			Description
57	−72,	16,	−12	−70,	14,	4	
58	−70,	14,	4	−69,	2,	16	
59	−69,	2,	16	−69,	−2,	16	Rear Engine Wall
60	−69,	−2,	16	−70,	−15,	4	
61	−70,	−15,	4	−72,	−16,	−12	
62	−56,	57,	−9	−56,	16,	−13	Left Wing Flap
63	−56,	−57,	−9	−56,	−16,	−13	Right Wing Flap
64	−93,	0,	48	−87,	0,	48	
65	−87,	0,	48	−70,	0,	23	Tail Flap
66	−70,	0,	23	−80,	0,	23	
67	−69,	15,	4	−69,	18,	12	
68	−69,	18,	12	−69,	10,	20	Rear Panel Of Left Engine Pod
69	−69,	10,	20	−69,	2,	16	
70	−69,	−15,	4	−69,	−19,	12	
71	−69,	−19,	12	−69,	−11,	20	Rear Panel Of Right Engine Pod
72	−69,	−11,	20	−69,	−2,	16	
73	−69,	15,	4	−44,	13,	5	
74	−69,	2,	16	−44,	2,	14	
75	−69,	18,	12	−52,	18,	13	
76	−69,	10,	20	−52,	10,	21	Left Engine Pod
77	−52,	18,	13	−52,	10,	21	
78	−52,	18,	13	−44,	13,	5	
79	−52,	10,	21	−44,	2,	14	
80	−69,	−15,	4	−44,	−13,	5	
81	−69,	−2,	16	−44,	−2,	14	
82	−69,	−19,	12	−52,	−19,	13	
83	−69,	−11,	20	−52,	−10,	21	Right Engine Pod
84	−52,	−19,	13	−52,	−10,	21	
85	−52,	−19,	13	−44,	−13,	5	
86	−52,	−10,	21	−44,	−2,	14	
87	−80,	0,	16	−81,	3,	14	
88	−81,	3,	14	−82,	3,	8	
89	−82,	3,	8	−83,	0,	6	Rear Opening Of Upper Engine
90	−83,	0,	6	−82,	−3,	8	
91	−82,	−3,	8	−81,	−3,	14	
92	−81,	−3,	14	−80,	0,	16	
93	−80,	0,	16	−69,	0,	12	
94	−83,	0,	6	−70,	0,	4	Body Of Upper Engine
95	−82,	3,	11	−69,	2,	8	
96	−82,	−3,	11	−69,	−2,	−8	
97	−69,	0,	12	−69,	2,	8	
98	−69,	2,	8	−70,	0,	4	Nozzle Of Upper Engine
99	−70,	0,	4	−69,	−2,	8	
100	−69,	−2,	8	−69,	0,	12	

References

Roy E. Meyers, *Microcomputer Graphics*, Addison-Wesley Micro Books. A very good reference on all types of graphics written for the Apple but easy to convert to TRS-80C.

Franklin C. Crow, "Three Dimensional Computer Graphics," *Byte*, March 1981.
Helpful background material and three-dimensional rotation.

John Ennis, "Rotational Grafx," *TRS-80 Microcomputer News*, June, 1982.
16K program somewhat like this one.

Roy Green, "Is a Rose in Color Still a Rose?," *80 Micro*, March, 1982.
16K program to rotate an object.

Variable List

A = 1 or 0 (screen page 5 or 1)
C1 = COS (THETA)
C2 = COS (PHI)
CD = Current Data
CX = Center (X)
CY = Center (Y)
D = Distance
I = Loop Counter
L = Loop Counter
LC = Line Count (number of lines)
P1 = PHI From
P2 = PHI To
P3 = PHI Difference and Step
PF$ = Picture File Name
PH = PHI
PI = The Value of PI
PL$ = Notes
Q = Temporary Variable
Q$ = Dummy Variable
R1 = RHO From
R2 = RHO To
R3 = RHO Difference and Step
RH = RHO
S1 = SIN (THETA)
S2 = SIN (PHI)
SA = Steps Allowed
SC = Scale
SP = Start of Permanent Data
ST = Number of Steps
SV = Start of Variable Data
T1 = THETA From
T2 = THETA To
T3 = THETA Difference and Step
TH = THETA
VP = Variable Poke
X1 = 3D Coordinates (From)
X2 = 3D Coordinates (To)
Y1 = 3D Coordinates (From)
Y2 = 3D Coordinates (To)
Z1 = 3D Coordinates (From)
Z2 = 3D Coordinates (To)
X3 = Calculated Coordinates
X4 = Calculated Coordinates
Y3 = Calculated Coordinates
Y4 = Calculated Coordinates
Z3 = Calculated Coordinates
Z4 = Calculated Coordinates
XA = Line From
XB = Line To
YA = Line From
YB = Line To

Line Number	From			To			Description
101	−83,	8,	2	−84,	13,	−1	
102	−84,	13,	−1	−85,	13,	−6	
103	−85,	13,	−6	−85,	8,	−9	Rear Opening Of Left Engine
104	−85,	8,	−9	−85,	3,	−6	
105	−85,	3,	−6	−84,	3,	−1	
106	−84,	3,	−1	−83,	8,	2	
107	−83,	8,	2	−72,	7,	−2	
108	−85,	8,	−9	−73,	7,	−9	
109	−84,	13,	−3	−73,	10,	−5	Body Of Left Engine
110	−84,	3,	−3	−73,	4,	−5	
111	−72,	7,	−2	−73,	10,	−5	
112	−73,	10,	−5	−73,	7,	−9	Nozzle Of Let Engine
113	−73,	7,	−9	−73,	4,	−5	
114	−73,	4,	−5	−72,	7,	−2	
115	−83,	−8,	2	−84,	−13,	−1	
116	−84,	−13,	−1	−85,	−13,	−6	
117	−85,	−13,	−6	−85,	−8,	−9	Rear Opening Of Right Engine
118	−85,	−8,	−9	−85,	−4,	−6	
119	−85,	−4,	−6	−84,	−4,	−1	
120	−84,	−4,	−1	−83,	−8,	2	
121	−83,	−8,	2	−72,	−7,	−2	
122	−85,	−8,	−9	−73,	−7,	−9	Body Of Right Engine
123	−84,	−13,	−3	−73,	−10,	−5	
124	−84,	−4,	−3	−73,	−4,	−5	
125	−72,	−7,	−2	−73,	−10,	−5	
126	−73,	−10,	−5	−73,	−7,	−9	Nozzle Of Right Engine
127	−73,	−7,	−9	−73,	−4,	−5	
128	−73,	−4,	−5	−72,	−7,	−2	
129	61,	2,	10	60,	5,	10	
130	60,	5,	10	57,	5,	13	Window — Forward Left
131	57,	5,	13	58,	3,	13	
132	58,	3,	13	61,	2,	10	
133	61,	−2,	10	60,	−5,	10	
134	60,	−5,	10	57,	−5,	13	Window — Forward Right
135	57,	−5,	13	58,	−3,	13	
136	58,	−3,	13	61,	−2,	10	
137	60,	5,	10	57,	9,	8	
138	57,	9,	8	55,	7,	12	Window — Middle Left
139	55,	7,	12	57,	5,	13	
140	60,	−5,	10	57,	−9,	8	
141	57,	−9,	8	55,	−7,	12	Window — Middle Right
142	55,	−7,	12	57,	−5,	13	
143	57,	9,	8	53,	10,	9	
144	53,	10,	9	53,	7,	12	Window — Left
145	53,	7,	12	55,	7,	12	
146	57,	−9,	8	53,	−10,	9	
147	53,	−10,	9	53,	−7,	12	Window — Right
148	53,	−7,	12	55,	−7,	12	
149	45,	13,	5	−45,	13,	4	Left Side Of Body
150	45,	−13,	5	−45,	−13,	4	Right Side Of Body
151	45,	13,	−12	48,	13,	−12	
152	48,	13,	−12	78,	6,	−9	Lower Left Side Of Nose
153	78,	6,	−9	85,	3,	−5	
154	85,	3,	−5	86,	0,	−4	
155	45,	−13,	−12	48,	−13,	−12	
156	48,	−13,	−12	78,	−7,	−9	Lower Right Side Of Nose
157	78,	−7	−9	85,	−3,	−5	
158	85,	−3	−5	86,	0,	−4	

For the first try, use a Distance 1 of 1200, Angle Around 1 of 0°, Elevation Angle 1 of 0°, Distance 2 of 1200, Angle Around 2 of 90°, an Elevation Angle 2 of 90°, and 3 steps.







☎ Enter 60 on TeleResponse page 159


```
100 FOR X=1 TO 1536 STEP 6
110 PRINT"FROM X,Y,Z"
120 INPUT X1,Y1,Z1:IF X1=999 THE
N 190
130 PRINT"TO X,Y,Z"
140 INPUT X2,Y2,Z2
150 GOSUB 930:IF E=1 THEN 110
160 A(X)=X1:A(X+1)=Y1:A(X+2)=Z1:
A(X+3)=X2:A(X+4)=Y2:A(X+5)=Z2
170 PRINT
180 NEXT X
190 LC=(X-1)/6
200 GOTO 50
210 'EDIT OLD PICTURE
220 CLS:PRINT"EDIT"
230 PRINT"CHANGE (1)":PRINT"DELE
TE (2)":PRINT"ADD (3)":PRINT"LIS
T (4)":PRINT"EXIT TO MAIN MENU (
5)";
240 INPUT Q:ON Q GOTO 260,330,44
0,520,50
250 GOTO 220
260 'CHANGE
270 PRINT:INPUT"WHICH LINE NUMBE
R";LN:LN=LN*6-5
280 PRINT:PRINTA(LN);A(LN+1);A(L
N+2):PRINTA(LN+3);A(LN+4);A(LN+5
)
290 PRINT:INPUT"FROM X,Y,Z";X1,Y
1,Z1:INPUT"TO X,Y,Z";X2,Y2,Z2
300 GOSUB 930:IF E=1 THEN 290
310 A(LN)=X1:A(LN+1)=Y1:A(LN+2)=
Z1:A(LN+3)=X2:A(LN+4)=Y2:A(LN+5)
=Z2
320 GOTO 210
330 'DELETE
340 PRINT:INPUT"WHICH LINE NUMBE
R TO DELETE";LN
350 LN=LN*6-5
360 PRINT:PRINTA(LN);A(LN+1);A(L
N+2):PRINTA(LN+3);A(LN+4);A(LN+5
)
370 INPUT"DELETE (Y OR N)";QA$
380 IF QA$<>"Y" THEN PRINT:PRINT
"NOT DELETED":GOTO 230
390 PRINT"WORKING..."
400 FOR X=LN TO LC*6-6
410 A(X)=A(X+6)
420 NEXT X
430 LC=LC-1:GOTO 210
440 'ADD
450 LN=LC*6+1
460 PRINT"COORDINATES TO ADD":IN
PUT"FROM X,Y,Z";X1,Y1,Z1
470 INPUT"TO X,Y,Z";X2,Y2,Z2
480 GOSUB 930:IF E=1 THEN 460
```

```
490 A(LN)=X1:A(LN+1)=Y1:A(LN+2)=
Z1:A(LN+3)=X2:A(LN+4)=Y2:A(LN+5)
=Z2
500 LC=LC+1
510 GOTO 210
520 'LIST
530 CLS:A=1:B=0
540 FOR X=A TO A+4*6-1 STEP 6
550 B=B+1
560 PRINTB,A(X);A(X+1);A(X+2)
570 PRINT"",A(X+3);A(X+4);A(X+5)

580 IF B=LC THEN 230
590 PRINT:NEXT X
600 PRINT:PRINT"HIT ENTER TO CON
TINUE"
610 IF INKEY$=""THEN 610
620 A=A+24
630 PRINT
640 GOTO 540
650 'SAVE
660 CLS:POKE65494,0
670 PRINT"WORKING...":POKE16384,
LC
680 FOR X=1 TO LC*6
690 POKE16384+X,A(X)+128
700 NEXT X
710 PRINT"DRAWING NAME (8 CHARAC
TERS OR   LESS)"
720 INPUT DN$
730 PRINT"TURN RECORDER TO RECOR
D AND HIT ENTER"
740 INPUT Q$:PRINT"SAVING..."
750 CSAVEM DN$,16384,16384+LC*6+
1,350
760 PRINT"DONE-TURN RECORDER OFF
"
770 PLAY"T20CDEFG"
780 INPUT"READY";Q$
790 GOTO 50
800 'LOAD
810 CLS:POKE65494,0
820 INPUT"DRAWING NAME";DN$
830 INPUT"TURN RECORDER TO PLAY
AND HIT   ENTER";Q$:PRINT"LOADIN
G..."
840 CLOADM DN$
850 PRINT"DONE-TURN RECORDER OFF
"
860 PLAY"T20CDEFG"
870 PRINT"WORKING...":LC=PEEK(16
384)
880 FOR X=1 TO LC*6
890 A(X)=PEEK(16384+X)-128
900 NEXT X
910 INPUT"READY";Q$
```

☎ Enter 62 on TeleResponse page 159


```
920 GOTO 50
930 E=0:IF X1<-128 OR X1>127 OR
Y1<-128 OR Y1>127 OR Z1<-128 OR
Z1>127 OR X2<-128 OR X2>127 OR Y
2<-128 OR Y2>127 OR Z2<-128 OR Z
2>127 THEN PRINT"ERROR-MUST BE I
N THE RANGE OF  -128 TO 127":E=
1
940 RETURN                    ■■■
```

Program Listing 2. Rotating Display

```
10 CLEAR10,16383:PCLEAR8 '3D ROT
ATING DISPLAY by Dale A. Keller
20 SP=16385:SV=17921:SC=1.09:PL$
="T20CDEFG"
30 '1536 BYTES FOR PERM DATA,148
47 BYTES FOR TEMP DATA
40 'LOAD
50 CLS:PRINT"LOADER":PRINT:INPUT
"PICTURE FILENAME";PF$
60 INPUT"SET RECORDER TO PLAY.
THEN HIT ENTER.";Q$:PRINT"LOADIN
G..."
70 POKE65494,0:CLOADM PF$:LC=PEE
K(SP-1)
80 PRINT"LOADED-TURN RECORDER OF
F":POKE65495,0:PLAYPL$
90 SA=INT((32768-SV)/4/LC):PRINT
:PRINT"YOU MAY HAVE UP TO"SA" ST
EPS"
100 INPUT"READY (HIT ENTER)";Q$
110 'INPUT COORDINATES
120 CLS:INPUT"DISTANCE 1, ANGLE
AROUND 1,    ANGLE ELEVATION 1"
;R1,T1,P1
130 PRINT:INPUT"DISTANCE 2, ANGL
E AROUND 2,    ANGLE ELEVATION
2";R2,T2,P2
140 PI=3.1416:T1=T1*PI/180:P1=P1
*PI/180:T2=T2*PI/180:P2=P2*PI/18
0
150 PRINT:INPUT"NUMBER OF STEPS"
;ST:IFST>SA THENPRINT"YOU MAY ON
LY HAVE";SA;" STEPS":GOTO150:ELS
EPRINT:PRINT"CALCULATING..."
160 'CALCULATE
170 R3=(R2-R1)/(ST-1):T3=(T2-T1)
/(ST-1):P3=(P2-P1)/(ST-1)
180 CX=128:CY=96:D=1000:RH=R1:TH
=T1:PH=P1:VP=SV
190 FOR I=1 TO ST
200 S1=SIN(TH):C1=COS(TH):S2=SIN
(PH):C2=COS(PH)
```

```
210 FORL=SP TO SP+(LC*6-1)STEP6
220 X1=PEEK(L)-128:Y1=PEEK(L+1)-
128:Z1=PEEK(L+2)-128:X2=PEEK(L+3
)-128:Y2=PEEK(L+4)-128:Z2=PEEK(L
+5)-128
230 X3=-X1*S1+Y1*C1:Y3=-X1*C1*C2
-Y1*S1*C2+Z1*S2:Z3=-X1*S2*C1-Y1*
S2*S1-Z1*C2+RH
240 X4=-X2*S1+Y2*C1:Y4=-X2*C1*C2
-Y2*S1*C2+Z2*S2:Z4=-X2*S2*C1-Y2*
S2*S1-Z2*C2+RH
250 XA=SC*(D*X3/Z3+CX):YA=CY-D*Y
3/Z3
260 XB=SC*(D*X4/Z4+CX):YB=CY-D*Y
4/Z4
270 XA=INT(XA):YA=INT(YA)
280 XB=INT(XB):YB=INT(YB)
290 POKE VP,XA:POKE VP+1,YA:POKE
 VP+2,XB:POKE VP+3,YB
300 VP=VP+4:NEXT L
310 RH=RH+R3:TH=TH+T3:PH=PH+P3
320 NEXT I
330 'DRAW SCREENS
340 PLAYPL$
350 PRINT:PRINT:INPUT"HIT ENTER
TO DISPLAY";Q$
360 PMODE4,5:PCLS:PMODE4,1:PCLS:
A=0:CD=SV
370 FOR I=1 TO ST
380 FOR L=1 TO LC
390 XA=PEEK(CD):YA=PEEK(CD+1):XB
=PEEK(CD+2):YB=PEEK(CD+3)

400 LINE(XA,YA)-(XB,YB),PSET
410 CD=CD+4:NEXT L
420 SCREEN1,0
430 IF A=1 THEN A=0 ELSE A=1
440 IF A=1 THEN PMODE 4,5:PCLS:E
LSE PMODE 4,1:PCLS
450 NEXT I
460 IFINKEY$=""THEN460
470 PRINT:PRINT"SAME RUN AGAIN (
1),":INPUT"SAME OBJECT, NEW ROTA
TION (2),  NEW OBJECT (3)";Q:ON
Q GOTO330,120,40
```

Program Listing 3. Scale Adjust

```
5 'SCALE ADJUST
10 PMODE 4,1: PCLS: SCREEN 1,1
20 LINE(50,50)-(150,150),PSET,B
30 GOTO 30
```